

Configuration of Two-Level Alarms

MultiLogger Application Note #11

Overview

This Application Note will provide instructions to configure two-level alarms using MultiLogger. The purpose of a two-level alarm is to allow a low-level alarm threshold to trigger one action and a high-level alarm to trigger a separate action.

Typically this is used in conjunction with the control ports of the control module, where each type of alarm will activate a respective control port, to presumably trigger an attached device such as an alarm light or alarm input to an attached system.

This Application Note will provide direction to activate control ports, in addition to a specialized port activation using the Visonic two-channel dialer.

Note: The software options described in this Application Note ship beginning with MultiLogger version 2.1.2, please contact Canary Systems or your software vendor to update your software. Software updates are available without charge to registered users via the Canary Systems website.


Software Configuration

There are two steps to the software configuration, the individual channels must be configured, then the Alarm Action must be configured.

The Visonic two-channel dialer configuration requires an additional software configuration step, as well as configuration of the dialer itself. See MultiLogger Application Note #8 for more information on the Visonic dialer.

Channel Configuration

The individual channels must be configured for the two-level alarm activation. See the screenshot to the right, note the **Two Level Alarm** option. The **Low Limit** value represents the first threshold value, the **High Limit** represents the second threshold value. Alarms WILL NOT be triggered together, i.e. if the High Limit is exceeded then the low alarm will not be activated, only the high alarm.



Alarm Action Configuration

Next the **Alarm Action** must be configured to activate the ports in case of alarm. The following options are presently supported in the Alarm Action list:

Two Level Alarm-C1&C2 - Port C1 will be activated for low alarm level, port C2 will be activated for the high alarm level.


Two Level Alarm-C5&C6 – Port C5 will be activated for low alarm level, port C6 will be activated for the high alarm level.

Visonic-C1C2 – Specialized port control to activate the two-channel Visonic dialer, see the following section for more information.

Three Lights - Port C1 will be activated for low alarm level, port C2 will be activated for the high alarm level, port C7 will be activated if neither alarm level is active.

Contact Canary Systems or your software vendor if your project has other requirements in terms of ports to activate and/or devices.

The screenshot below illustrates the **Two Level Alarm-C1&C2** option selected.



Software Configuration and Operation Details When Using Visonic-C1C2


The **Visonic-C1C2** Alarm Action includes additional functionality, besides activation of the control ports to activate each of the dialer inputs, to enhance dialer functionality.

This functionality includes:

- The ability to specify a number of consecutive alarms before the dialer is activated. The default is 6, this can be changed by modifying the instruction file. This prevents dialer activation due to spurious readings. Press the gear button with Visonic-C1C2 selected as the Alarm Action to load the instruction file into the supplied editor. See the comments in the instruction for changing the number of consecutive alarms. Save your changes when finished.
- The ability to disable the dialer after alarm activation to prevent re-activation of the dialer. Flag 1 is used to disable the low alarm level, Flag 2 is used to disable the high alarm level. After an alarm is activated you must run MultiLogger, initiate the Monitor mode with the system and click on the appropriate flag to reset it LOW. **FURTHER ALARMS WILL NOT BE ACTIVATED UNTIL THIS IS DONE.**


Note! To use the **Visonic-C1C2** Alarm Action you will need to configure 2 locations to be used for counting each type of alarm. Typically Input Locations 19 and 20 are used for this purpose, they will need to be named as **AlarmCount1** and **AlarmCount2**. The Update procedure will not succeed until this configuration has been done.

To configure the Input Locations, select the menu item **Program | Input Locations** in the Logger form. You will see all of the available Input Locations listed on the left side of the form, scroll down to location 19. Enter **AlarmCount1** as the Label, as shown. Do not modify any of the other options.



The screenshot shows the 'Configure Input Locations' dialog box with location 19 selected. The 'Location' list on the left has 19 highlighted. The 'Store Location' checkbox is unchecked. The 'Label' field contains 'AlarmCount1', 'Description' is 'User Location 1', 'Units' is 'None', and 'Reserved' is 'UserLoc1'. The 'Check Alarms' section has a gear icon, a lightbulb icon, and a 'Type' dropdown set to 'None'. Below the dropdown are 'Low Limit' and 'High Limit' fields, both set to 0. At the bottom are 'Help', 'Accept', and 'Cancel' buttons.

Select location 20 and configure the Label as **AlarmCount2**, as shown. Do not modify any of the other options.



The screenshot shows the 'Configure Input Locations' dialog box with location 20 selected. The 'Location' list on the left has 20 highlighted. The 'Store Location' checkbox is unchecked. The 'Label' field contains 'AlarmCount2', 'Description' is 'User Location 2', 'Units' is 'None', and 'Reserved' is 'UserLoc2'. The 'Check Alarms' section has a gear icon, a lightbulb icon, and a 'Type' dropdown set to 'None'. Below the dropdown are 'Low Limit' and 'High Limit' fields, both set to 0. At the bottom are 'Help', 'Accept', and 'Cancel' buttons.

Two Level Alarm-C1&C2 Instruction File Listing

*;This instruction file will set Port 1 HIGH if alarm flag 6 is HIGH, otherwise Port 1 is set low
;It will also set Port 2 HIGH if alarm flag 8 is HIGH, otherwise Port 2 is set low
;Remember flag 7 is used to enable/disable alarm activation!!!
;Check flag 6 for alarm level 1*

```
P20   Set Port(s)      ;
1:[9999           ]    C8,C7,C6,C5 Options      ;
2:[9977           ]    C4,C3,C2,C1 Options      ;

P91   If Flag/Port    ;
1:[16            ]    Flag/Port Options (Do if Flag 6 is High)      ;
2:[30            ]    Command Code Option (Then Do) ;

P86   Do              ;
1:[41            ]    Command Code Option (Set Port 1 High) ;

P94   Else            ;

P86   Do              ;
1:[51            ]    Command Code Option (Set Port 1 Low) ;

P95   End             ;
```

;check flag 8 for alarm level 2

```
P91   If Flag/Port    ;
1:[18            ]    Flag/Port Options (Do if Flag 8 is High)      ;
2:[30            ]    Command Code Option (Then Do) ;

P86   Do              ;
1:[42            ]    Command Code Option (Set Port 2 High) ;

P94   Else            ;

P86   Do              ;
1:[52            ]    Command Code Option (Set Port 2 Low) ;

P95   End             ;
```

*;We need to reset alarm flag 6 here because multilogger does not reset it automatically before
the next readings*

;Flag 8 IS reset automatically by Multilogger

```
P86   Do              ;
1:[26            ]    Command Code Option (Set Flag 6 Low) ;
```

Visonic-C1C2 Instruction File Listing

Note the instructions which allow you to specify the number of consecutive alarms before alarm activation will occur.

NOTE: For users writing their own programs avoid use of Flags 1, 2, 6, 7 and 8.

```
P20    Set Port(s)      ;
1:[9999          ]      C8,C7,C6,C5 Options   ;
2:[9977          ]      C4,C3,C2,C1 Options   ;

P86    Do              ;Do Instruction
01:[51          ]      Command Code Option (Set Port 1 Low)      ;reset the alarm

P86    Do              ;
1:[52          ]      Command Code Option (Set Port 2 Low) ;reset the alarm

;Check our Low Alarm Level flag
P91    If Flag/Port    ;Check flag 6
01:[16          ]      Flag/Port Options (Do if Flag 6 is High) ;if high (alarm
enabled)
02:[30          ]      Command Code Option (Then Do)      ;then do

P91    If Flag/Port    ;
1:[21          ]      Flag/Port Options (Do if Flag 1 is Low)    ;
2:[30          ]      Command Code Option (Then Do) ;

P32    Z=Z+1          ;
1:[AlarmCount1  ]      Z Loc ;

P89    If (X<=>F)      ;
1:[AlarmCount1  ]      X Loc ;
2:[3            ]      Comparison Code Option (>=) ;
3:[6            ]      F ; Set consecutive alarms here!!!
4:[30           ]      Command Code Option (Then Do) ;

P86    Do              ;
1:[11          ]      Command Code Option (Set Flag 1 High) ;

P86    Do              ;
1:[41          ]      Command Code Option (Set Port 1 High) ;

P22    Excitation with Delay ;
1:[1          ]      Ex Channel ;
2:[0          ]      Delay W/Ex (units = 0.01 sec) ;
3:[200        ]      Delay After Ex (units = 0.01 sec) ;
4:[0          ]      mV Excitation ;

P86    Do              ;
1:[51          ]      Command Code Option (Set Port 1 Low) ;

P30    Z=F            ;
1:[0          ]      F ;
2:[0          ]      Exponent of 10 ;
3:[AlarmCount1]      Z Loc ;

P95    End            ;
P95    End            ;
P94    Else            ;

;Reset alarms if alarm is low - ensures consecutive alarms
P30    Z=F            ;
1:[0          ]      F ;
2:[0          ]      Exponent of 10 ;
3:[AlarmCount1]      Z Loc ;
```

```

P95     End      ;

;Check our High Alarm Level Flag
P91     If Flag/Port      ;Check flag 8
01:[18          ]      Flag/Port Options (Do if Flag 8 is High)      ;if high (alarm
enabled)
02:[30          ]      Command Code Option (Then Do)      ;then do

P91     If Flag/Port      ;
1:[22          ]      Flag/Port Options (Do if Flag 2 is Low)      ;
2:[30          ]      Command Code Option (Then Do) ;

P32     Z=Z+1      ;
1:[AlarmCount2  ]      Z Loc      ;

P89     If (X<=>F)      ;
1:[AlarmCount2  ]      X Loc      ;
2:[3            ]      Comparison Code Option (>=)      ;
3:[6            ]      F      ; Set consecutive alarms here!!!!
4:[30           ]      Command Code Option (Then Do) ;

P86     Do      ;
1:[12          ]      Command Code Option (Set Flag 2 High) ;

P86     Do      ;
1:[42          ]      Command Code Option (Set Port 2 High) ;

P22     Excitation with Delay ;
1:[1          ]      Ex Channel      ;
2:[0          ]      Delay W/Ex (units = 0.01 sec) ;
3:[200        ]      Delay After Ex (units = 0.01 sec) ;
4:[0          ]      mV Excitation ;

P86     Do      ;
1:[52          ]      Command Code Option (Set Port 2 Low) ;

P30     Z=F      ;
1:[0          ]      F      ;
2:[0          ]      Exponent of 10 ;
3:[AlarmCount2]      Z Loc      ;

P95     End      ;

P95     End      ;

P94     Else      ;

;Reset alarms if alarm is low - ensures consecutive alarms
P30     Z=F      ;
1:[0          ]      F      ;
2:[0          ]      Exponent of 10 ;
3:[AlarmCount2]      Z Loc      ;

P95     End      ;

;Make sure we reset the Alarm Flag 6 - MultiLogger will automatically reset Alarm Flag 8
P86     Do      ;
1:[26          ]      Command Code Option (Set Flag 6 Low) ;

```